

Bacharelado em Sistemas de Informação
SIF001 – Algoritmos. Aula 11 – 19/05/2020
Profs. Antonello e Antero

EXERCÍCIOS

- 1) Desenvolver um algoritmo para ler 5 valores numéricos inteiros e armazene em um vetor.

Em seguida deve ser preenchido um segundo vetor, com os mesmo valores do primeiro, só que em ordem inversa. O primeiro elemento de um vetor será o último do outro vetor. O segundo elemento de um será o penúltimo do outro e assim por diante.

```
algoritmo "Ex01Vetores"

var
    A, B: vetor[1..5] de inteiro
    ind, x: inteiro
inicio
    escreval("Carregar valores no primeiro vetor")
    para ind de 1 ate 5 passo 1 faca
        escreva("Digite o ", ind, "º valor: ")
        leia(A[ind])
    fimpara

    // rotina que preenche o vetor B
    x <- 5
    para ind de 1 ate 5 passo 1 faca
        B[x] <- A[ind]
        x <- x-1
    fimpara

    escreval()
    escreval("Conteúdo do vetor A")
    para ind de 1 ate 5 passo 1 faca
        escreva(A[ind], " ")
    fimpara

    escreval()
    escreval("Conteúdo do vetor B")
    para ind de 1 ate 5 passo 1 faca
        escreva(B[ind], " ")
    fimpara
fimalgoritmo
```

- 2) Desenvolver um algoritmo para ler 10 valores numéricos do tipo real. Como resultado do processamento exiba quantos e liste quais elementos estão acima da média calculada entre eles.

```
algoritmo "Ex02AcimaMedia"

var
    vet: vetor[1..10] de real
    media, soma: real
    quantos, i: inteiro

inicio
    escreval("Digitar 10 valores para o vetor")
    soma <- 0
    para i de 1 ate 10 passo 1 faca
        leia(vet[i])
        soma <- soma + vet[i]
    fimpara

    // cálculo da média
    media <- soma / 10

    // encontra lista valores acima da média
    escreval()
    escreval("Média: ", media)
    escreval("Lista de valores acima da média")
    quantos <- 0
    para i de 1 ate 10 passo 1 faca
        se (vet[i] > media) entao
            escreva(vet[i], " ")
            quantos <- quantos + 1
        fimse
    fimpara
    escreval()
    escreval("Total de valores acima da média: ", quantos)
fimalgoritmo
```

- 3) Declarar três variáveis indexadas A, B e C com capacidade para armazenar, cada uma, cinco elementos do tipo inteiro. Solicitar a entrada de dados para os elementos de A e B. Fazer cada elemento de C igual aos respectivos elementos de A multiplicado por B. Exibir todos os valores de A, B e C.

```
algoritmo "Aula13_Ex01"
var
  A, B, C: Vetor[1..5] de inteiro
  ind: inteiro

inicio

// *****
// *** PRIMEIRA SOLUÇÃO ***
// *****
Escreval("ROTINA DE CARGA DE DADOS NO VETOR A")
Para ind de 1 Ate 5 Passo 1 Faca
  Escreva("Informe o ", ind, " elemento do vetor A: ")
  Leia (A[ind])
FimPara

Escreval("ROTINA DE CARGA DE DADOS NO VETOR B")
Para ind de 1 Ate 5 Passo 1 Faca
  Escreva("Informe o ", ind, " elemento do vetor B: ")
  Leia (B[ind])
FimPara

// Rotina que grava valores no vetor C (ElemC = ElemA * ElemB)
Para ind de 1 Ate 5 Passo 1 Faca
  C[ind] <- A[ind] * B[ind]
FimPara

Escreval("ROTINA QUE MOSTRA VALORES ARMAZENADOS NOS VETORES")
Escreval("Vetor A":10, "Vetor B":10, "Vetor C":10)
Para ind de 1 Ate 5 Passo 1 Faca
  Escreval(A[ind]:4, B[ind]:10, C[ind]:10)
FimPara

// *****
// *** SEGUNDA SOLUÇÃO ***
// *****
Escreval("ROTINA DE CARGA DE DADOS NO VETOR A")
Para ind de 1 Ate 5 Passo 1 Faca
  Escreva("Informe o ", ind, " elemento do vetor A: ")
  Leia (A[ind])
FimPara

Escreval("ROTINA DE CARGA DE DADOS NO VETOR B")
Para ind de 1 Ate 5 Passo 1 Faca
  Escreva("Informe o ", ind, " elemento do vetor B: ")
  Leia (B[ind])
  // grava valores no vetor C (ElemC = ElemA * ElemB)
  C[ind] <- A[ind] * B[ind]
FimPara

Escreval("ROTINA QUE MOSTRA VALORES ARMAZENADOS NOS VETORES")
Escreval("Vetor A":10, "Vetor B":10, "Vetor C":10)
Para ind de 1 Ate 5 Passo 1 Faca
  Escreval(A[ind]:4, B[ind]:10, C[ind]:10)
```

FimPara

```
// *****  
// *** TERCEIRA SOLUÇÃO ***  
// *****  
Escreval("ROTINA DE CARGA DE DADOS NOS VETORES A E B")  
Para ind de 1 Ate 5 Passo 1 Faca  
    Escreva("Informe o ", ind, " elemento do vetor A: ")  
    Leia (A[ind])  
    Escreva("Informe o ", ind, " elemento do vetor B: ")  
    Leia (B[ind])  
    // grava valores no vetor C (ElemC = ElemA * ElemB)  
    C[ind] <- A[ind] * B[ind]  
FimPara  
  
Escreval("ROTINA QUE MOSTRA VALORES ARMAZENADOS NOS VETORES")  
Escreval("Vetor A":10, "Vetor B":10, "Vetor C":10)  
Para ind de 1 Ate 5 Passo 1 Faca  
    Escreval(A[ind]:4, B[ind]:10, C[ind]:10)  
FimPara
```

Fimalgoritmo

- 4) Desenvolver um algoritmo que possibilite armazenar em uma variável indexada sete valores numéricos do tipo inteiro. Determinar qual é o menor valor armazenado e indicar qual posição da variável ele ocupa.

```
algoritmo "Aula13_Ex02"
var
    Vet: Vetor [1..7] de inteiro
    ind, Men, PosiMen: inteiro

inicio

// Rotina que armazenar valores no vetor
// entrada de dados por digitação no teclado durante a
// execução do programa
Escreval("ROTINA DE CARGA DE DADOS NO VETOR")
Para ind de 1 Ate 7 Passo 1 Faca
    Escreva("Informe o ", ind, " elemento do vetor: ")
    Leia (Vet[ind])
FimPara

// Rotina que encontra o menor valor armazenado no vetor
// e mostra em que posição ele se encontra
// Se ocorrer mais de um, apenas o primeiro elemento será mostrado

PosiMen <- 1
Para ind de 2 Ate 7 Passo 1 Faca
    Se (Vet[ind] < Vet[PosiMen]) Entao
        PosiMen <- ind
    FimSe
FimPara
Escreval()
Escreval("Menor elemento do vetor: ", Vet[PosiMen])
Escreval("Posição do menor elemento: ", PosiMen)

Fimalgoritmo
```